Лабораторная работа 4.

Теоретические положения.

Бинарное дерево – это конечное множество узлов, которое либо является пустым, либо состоит из корня и двух непересекающихся бинарных деревьев, которые называются левым и правым поддеревьями данного корня.

Примеры.

В качестве примеров приведем рассмотренные в лекции процедуры и функции (по всей вероятности, это напоминание не будет лишним).

Объявление типа:

type 

  T_item = integer; // здесь может быть любой тип

  P_Node = ^T_Node;

  T_Node = record

             info: T_item;

             key: integer;

             left, right: P_Node;

           end;

var root: P_Node; // корень дерева

Процедуры обхода дерева:

// вспомогательная процедура

procedure PrintKey(t: P_Node);

begin

  write(t^.key: 5);

end;

// обход сверху вниз

procedure preorder(t: P_Node);

begin

  if t <> nil 

    then begin

      PrintKey(t);

      preorder(t^.left);

      preorder(t^.right);

    end;

end;

// обход снизу вверх

procedure postorder(t: P_Node);

begin

  if t <> nil 

    then begin

      postorder(t^.left);

      postorder(t^.right);

      PrintKey(t);

    end;

end;

// обход слева направо

procedure inorder(t: P_Node);

begin

  if t <> nil 

    then begin

      inorder(t^.left);

      PrintKey(t);

      inorder(t^.right);

    end;

end;

// вставка элемента в дерево (рекурсивная процедура)

procedure InsertInTree(var t: P_Node; k: Integer);

begin

  if t = nil // если найдено место для нового узла

    then begin

           new(t); // распределяем память

           t^.left := nil; // устанавливаем пустые ссылки для левого и

           t^.right := nil; // правого поддеревьев

           t^.key := k; // заполняем ключевое поле

           {а здесь не помешало бы заполнить информационное поле}

         end // then 

{иначе определяем, в каком поддереве узла t должен располагаться элемент с таким ключом}

    else if k <= t^.key // если новый ключ меньше или равен текущему

           then InsertInTree(t^.left, k) // вставим в левое поддерево

           else InsertInTree(t^.right, k); // а иначе – в правое

end;

// процедура удаления элемента (по Вирту)

procedure Delete_Tree(var t: P_Node; k: Integer);

var q: P_Node;

procedure Del(var w: P_Node); // поиск самого правого элемента

begin

  if w^.right <> nil then Del(w^.right)

                     else begin

                            q := w;

 // запоминаем адрес, чтобы потом удалить этот элемент

                            t^.key := w^.key;

                            w := w^.left;

                     end;

end;

begin

  if t <> nil then

    if k < t^.key then Delete_Tree(t^.left, k)

    else if k > t^.key then Delete_Tree(t^.right, k)

         else begin

           q := t;

           if t^.right = nil then t := t^.left 

           // правого поддерева нет

           else if t^.left = nil then t := t^.right

          // левого поддерева нет

                else Del(t^.left);

               // находим самый правый элемент в левом поддереве

           dispose(q);

         end;

end;

Задачи.

1. Описать логическую функцию same(T), определяющую, есть ли в дереве T хотя бы два одинаковых элемента.

2. Написать процедуру удаления (из существующего бинарного дерева) всех отрицательных элементов.

3. Описать рекурсивную функцию или процедуру, которая определяет, входит ли элемент Е в дерево Т.

4. Описать рекурсивную функцию или процедуру, которая определяет число вхождений элемента Е в дерево Т.

5. Описать рекурсивную функцию или процедуру, которая вычисляет сумму элементов непустого дерева.

6. Описать рекурсивную функцию или процедуру, которая находит величину наибольшего элемента непустого дерева.

7. Описать рекурсивную функцию или процедуру, которая печатает элементы из всех листьев дерева.

8. Описать рекурсивную функцию или процедуру, которая определяет максимальную глубину непустого дерева Т, т.е. Число ветвей в самом длинном из путей от корня дерева до листьев.

9. Описать рекурсивную функцию или процедуру, которая подсчитывает число вершин на n-ом уровне непустого дерева Т(корень считать вершиной 0-го уровня).

